

This document provides a demo that shows how our scripts perform. The examples are obtained from section **5.1 The linear model** and section **8.9 Seasonal ARIMA models**. You can find the full scripts in the **Scripts** folder of **Forecasting-Cookbook** repository at <https://github.com/sassoftware/sas-viya-forecasting/tree/master/Forecasting-Cookbook>.

Before running the scripts, you need to define a library (time) that contains your SAS datasets. Moreover, you need to set up a CAS session and link a library (mycas) to it.


5.1 The linear model

```
1  /**** Chapter 5 Time series regression models ****/  
2  /**** 5.1 The linear model ****/  
3  /**** https://otexts.org/fpp2/regression-intro.html ****/  
4  
5  *sending data from standard SAS to CAS;  
6  data mycas.fpp2_uschange;  
7      set time.fpp2_uschange;  
8  run;  
9  
10 /**** Simple linear regression ****/  
11 /**** Example: US consumption expenditure ****/  
12 proc regselect data=mycas.fpp2_uschange;  
13     model consumption = income;  
14     *output upper (UCL) and lower (LCL) bound limit for fitted values;  
15     output out=mycas.fpp2_uschange_pred predicted ucl lcl  
16     copyvars = (consumption income);  
17 run;  
18  
19 proc sgplot data = mycas.fpp2_uschange_pred;  
20     series x = income y=Pred;  
21     scatter x = income y=consumption;  
22 run;
```

Parameter Estimates, and its relevant statistics are displayed in the output:

Parameter Estimates					
Parameter	DF	Estimate	Standard Error	t Value	Pr > t
Intercept	1	0.545104	0.055687	9.79	<.0001
Income	1	0.280601	0.047442	5.91	<.0001

The following is the plot of the fitted values and actual values of consumption with respect to the income levels.



8.9 Seasonal ARIMA models

The automatic time series analysis and forecasting (ATSM) package provides objects that are designed to support automatic time series modeling and automatic forecasting. For more information about the statistical methodology that underlies this package, see the chapters about the HPFDIAGNOSE, HPFENGINE, and HPFSELECT procedures in *SAS Forecast Server Procedures: User's Guide*. In addition, it is helpful to review [Chapter 10: The TSMODEL Procedure](#) in SAS Visual Forecasting 8.4: Forecasting Procedures, and [Chapter 12: Time Series Model Package](#). Each of the objects in the ATSM package is designed to carry out a particular task in the time series analysis process.

```

1  /**** 8.9 Seasonal ARIMA models ****/
2  /**** Example: Corticosteroid drug sales in Australia ****/
3  /**** https://otexts.com/fpp2/seasonal-arima.html ****/
4
5  data mycas.fpp2_h02;
6      set time.fpp2_h02;
7  run;
8
9
10 *automatical diagnose the time seires to get candidate ARIMA models;
11 proc tsmodel data      = mycas.fpp2_h02
12     outobj      = (outest = mycas.outest
13                   outfor = mycas.outfor
14                   outstat = mycas.outSummary);
15     id date interval = month;
16     var H02;
17     require atsm;
18     submit;
19     declare object diagnose(diagnose);
20     declare object diagspec(diagspec);
21     declare object dataframe(tsdf);
22     declare object forecast(foreng);
23     declare object outest(outest);
24     declare object outfor(outfor);
25     declare object outstat(outstat);
26
27     *specify dataframe information;
28     rc = dataframe.initialize();
29     rc = dataframe.addy(H02);
30

```

```

31 *set diagnose parameter;
32 rc = diagspec.open();
33 rc = diagspec.setarimax('estmethod','ml'); *set arima models to be considered in diagnose
34 rc = diagspec.close();
35
36 *run diagnose;
37 rc = diagnose.initialize(dataframe);
38 rc = diagnose.setspec(diagspec);
39 rc = diagnose.setoption('holdout', 12);
40 rc = diagnose.setoption('back', 24);
41 rc = diagnose.run();
42
43 *run forecast engine;
44 rc = forecast.initialize(diagnose);
45 rc = forecast.setoption('criterion','rmse');
46 rc = forecast.setoption('holdout', 12);
47 rc = forecast.setoption('back', 24);
48 rc = forecast.setoption('lead', 24);
49 rc = forecast.run();
50
51 *collect output;
52 rc = outest.collect(forecast);
53 rc = outfor.collect(forecast);
54 rc = outstat.collect(forecast);
55 endsubmit;
56 quit;

```

In this example, the best ARIMA model is fitted on the data. One can create a plot of fitted values as below:

```

58 proc sgplot data = mycas.outfor;
59   series x = date y = actual;
60   scatter x = date y = predict;
61 run;

```

